Euler-Heisenberg matching at one loop

Julian Eicher (julian.eicher@unibe.ch), April 2024

Preliminaries

- Load Package-X

In[ ]:=
```
<< X`
```

```
Package-X v2.1.1 [patched 22/08/2020], by Hiren H. Patel
For more information, see the guide
```

- The package can be downloaded here (folder “X”): https://gitlab.com/mule-tools/package-x

- Documentation: https://mule-tools.gitlab.io/package-x/downloads/primer.pdf

- Define kinematics

In[ ]:=
```
kinematics = {p1.p1 → 0, p2.p2 → 0, p1.p2 → s / 2}
```

Out[ ]= $\left\{p1.p1 \to 0,\ p2.p2 \to 0,\ p1.p2 \to \frac{s}{2}\right\}$

- Global prefactor (factor (-1) from the fermion trace)

In[ ]:=
```
prefactor = -e ^ 4;
```

- In the following, we calculate the full contracted forward amplitudes $A_1 = g_{\mu_1 \mu_2}\, g_{\nu_1 \nu_2}\, B^{\mu_1 \mu_2 \nu_1 \nu_2}$ and

$A_2 = g_{\mu_1 \nu_1} g_{\mu_2 \nu_2} B^{\mu_1 \mu_2 \nu_1 \nu_2}$ for each of the three diagrams separately, and then combine the results and expand in s

Diagram 1

In[ ]:=



- Calculate the Dirac trace in the numerator

In[∘]:= num1A1 =
$g_{\mu 1,\mu 2} * g_{\nu 1,\nu 2} *$ Spur[γ.k + m 𝟙, $\gamma_{\mu 1}$, γ.(k - p1) + m 𝟙,
$\gamma_{\nu 1}$, γ.k + m 𝟙, $\gamma_{\nu 2}$, γ.(k + p2) + m 𝟙, $\gamma_{\mu 2}$] // Contract

Out[∘]= $4 d^2 m^4 + 16 m^2 k.k - 16 d m^2 k.k +$
$4 d^2 m^2 k.k + 16 d m^2 (k.k - k.p1) -$
$8 d^2 m^2 (k.k - k.p1) + 16 d m^2 (k.k + k.p2) -$
$8 d^2 m^2 (k.k + k.p2) + 32 (k.k - k.p1) (k.k + k.p2) -$
$32 d (k.k - k.p1) (k.k + k.p2) +$
$8 d^2 (k.k - k.p1) (k.k + k.p2) +$
$4 d^2 m^2 (k.k - k.p1 + k.p2 - p1.p2) -$
$16 k.k (k.k - k.p1 + k.p2 - p1.p2) +$
$16 d k.k (k.k - k.p1 + k.p2 - p1.p2) -$
$4 d^2 k.k (k.k - k.p1 + k.p2 - p1.p2)$

In[∘]:= num1A2 =
$g_{\mu 1,\nu 1} * g_{\mu 2,\nu 2} *$ Spur[γ.k + m 𝟙, $\gamma_{\mu 1}$, γ.(k - p1) + m 𝟙,
$\gamma_{\nu 1}$, γ.k + m 𝟙, $\gamma_{\nu 2}$, γ.(k + p2) + m 𝟙, $\gamma_{\mu 2}$] // Contract

Out[∘]= $4 d^2 m^4 + 4 d^2 m^2 k.k + 16 d m^2 (k.k - k.p1) -$
$8 d^2 m^2 (k.k - k.p1) + 16 d m^2 (k.k + k.p2) -$
$8 d^2 m^2 (k.k + k.p2) + 32 (k.k - k.p1) (k.k + k.p2) -$
$32 d (k.k - k.p1) (k.k + k.p2) +$
$8 d^2 (k.k - k.p1) (k.k + k.p2) +$
$16 m^2 (k.k - k.p1 + k.p2 - p1.p2) -$
$16 d m^2 (k.k - k.p1 + k.p2 - p1.p2) +$
$4 d^2 m^2 (k.k - k.p1 + k.p2 - p1.p2) -$
$16 k.k (k.k - k.p1 + k.p2 - p1.p2) +$
$16 d k.k (k.k - k.p1 + k.p2 - p1.p2) -$
$4 d^2 k.k (k.k - k.p1 + k.p2 - p1.p2)$

Integrate

In[∘]:=
```
int1A1 =
  LoopIntegrate[num1A1, k, {k, m}, {k - p1, m},
    {k, m}, {k + p2, m}] /. kinematics;
```

In[∘]:=
```
int1A2 =
  LoopIntegrate[num1A2, k, {k, m}, {k - p1, m},
    {k, m}, {k + p2, m}] /. kinematics;
```

- The full results read

In[∘]:=
```
res1A1 = LoopRefine[int1A1] // DiscExpand
```

Out[∘]=

$$\frac{8\left(8 m^2-9 s\right)}{s}+16\left(\frac{1}{\epsilon}+\text{Log}\left[\frac{\mu^2}{m^2}\right]\right)+\frac{4\left(-4 m^2+s\right) \text{Log}\left[\frac{2 m^2-s+\sqrt{-\left(\left(4 m^2-s\right) s\right)}}{2 m^2}\right]^2}{s}+\frac{16\left(2 m^2-s\right) \sqrt{s\left(-4 m^2+s\right)}\ \text{Log}\left[\frac{2 m^2-s+\sqrt{s\left(-4 m^2+s\right)}}{2 m^2}\right]}{s^2}$$

In[∘]:=
```
res1A2 = LoopRefine[int1A2] // DiscExpand
```

Out[∘]=

$$\frac{64 m^2}{s}+16\left(\frac{1}{\epsilon}+\text{Log}\left[\frac{\mu^2}{m^2}\right]\right)+\frac{4\left(2 m^2+s\right) \text{Log}\left[\frac{2 m^2-s+\sqrt{-\left(\left(4 m^2-s\right) s\right)}}{2 m^2}\right]^2}{s}+\frac{8 \sqrt{s\left(-4 m^2+s\right)}\ \left(4 m^2+s\right) \text{Log}\left[\frac{2 m^2-s+\sqrt{s\left(-4 m^2+s\right)}}{2 m^2}\right]}{s^2}$$

- Expand for small *s*

In[∘]:=
```
Normal@Series[res1A1, {s, 0, 2}]
```

Out[∘]=

$$-\frac{56}{3}-\frac{24 s}{5 m^2}-\frac{109 s^2}{315 m^4}+16\left(\frac{1}{\epsilon}+\text{Log}\left[\frac{\mu^2}{m^2}\right]\right)$$

In[∘]:=
```
Normal@Series[res1A2, {s, 0, 2}]
```

Out[∘]= $-\frac{56}{3}-\frac{14\,s}{5\,m^2}-\frac{67\,s^2}{315\,m^4}+16\left(\frac{1}{\epsilon}+\mathrm{Log}\left[\frac{\mu^2}{m^2}\right]\right)$

Diagram 2

Diagram 3

Combine

- Combine and expand

In[∘]:=
```
A1full = res1A1 + res2A1 + res3A1;
A1exp = Series[A1full, {s, 0, 2}] // Normal
```

Out[∘]= $-\frac{19\,s^2}{45\,m^4}$

In[∘]:=
```
A2full = res1A2 + res2A2 + res3A2;
A2exp = Series[A2full, {s, 0, 2}] // Normal
```

Out[∘]= $-\frac{22\,s^2}{45\,m^4}$

- Multiply by the prefactor and include a factor of 2 for identical diagrams (also restore the factor I/(16*Pi^2) which Package-X removes)

In[∘]:=
```
A1final = I / (16 * Pi ^ 2) * prefactor * 2 * A1exp /.
   e → Sqrt[4 * Pi * α]
```

Out[∘]= $\frac{38\,i\,s^2\,\alpha^2}{45\,m^4}$

In[ ]:=
```
A2final = I / (16 * Pi ^ 2) * prefactor * 2 * A2exp /.
   e → Sqrt[4 * Pi * α]
```

Out[ ]=
$$\frac{44\, i\, s^2\, \alpha^2}{45\, m^4}$$